Practical Probability with Spreadsheets

Chapter 7: Time Series Models (very incomplete)

1. Log-Brownian motion models in finance

Brownian motion is named after the botanist Robert Brown (1773-1858), who looked through a microscope and saw particles of dust moving randomly around in water. Brownian motion is a mathematical model developed to describe the movement of such particles. Mathematicians understand Brownian motion as the limit of simple random walk models where, in each short interval of time, the particle makes a small step either to the left or to the right, so that the numerical value that represents the particle's position will either increase or decrease by some small amount.

In a Brownian-motion model, we have some process randomly generating, at each point in time, a value that we may interpret as measuring the position of some particle which is wandering along a line. During any interval of time, the difference between the final value and the initial value is the Brownian motion over that time interval. The Brownian motions over disjoint time intervals are independent random variables. As the sum of many small steps, the Brownian motion over any time interval will be a Normal random variable. The mean of the Brownian motion is proportional to the length of the time interval, but the standard deviation is proportional to the square-root of the length of the time interval, and the constants of proportionality are called the drift and the volatility respectively. That is, if the drift is D and the volatility is V then, during any time interval of length T (measured in years or some other time unit), the net change in the particle's position will be a Normal random variable with mean μ(T) and standard deviation σ(T) that satisfy the equations

$$\mu(T) = D*T$$

$$\sigma(T) = V*(T\hat{}0.5).$$

Figure 1 illustrates the calculations for a simple Brownian-motion model. In this case, let us say that time is being measured in years. The drift (0.18) and the volatility (0.34) of the Brownian motion are entered into cells A2 and A3 respectively. Thus, over any one-year interval, the Brownian motion will be a Normal random variable with mean 0.18 and standard deviation

0.34. Now suppose that we are interested in the Brownian motion over some other time interval, say 2 years, which has been entered into cell A4. The Brownian motion over this time interval is a Normal random variable with mean A2*A4 = 0.18*2 = 0.36 and standard deviation A3*(A4^.5) = 0.34*(2^.5) = 0.4808, which are computed in cells A5 and A6 respectively. If the initial value or position of the particle is 4, then its position after 2 years will be a Normal random variable with mean 4.36 and standard deviation 0.4808. With this intial value entered in cell A7, the positional value after T years is simulated in cell A8 by the formula

=A7+NORMINV(RAND(),A5,A6)

	A	B	C	D	E	F	G
1	Brownian motion over time T						
2	0.18	Drift					
3	0.34	Volatility					
4	2	Length of time interval T					
5	0.36	Mean of change over time T					
6	0.4808	Stdev of change over time T					
7	4	Initial value					
8	4.347	Simulated value, time T later					
9							
10	Log-Brownian motion over time T (with log-drift and						
11	log-volatility equal to the Drift and Volatility above)						
12	54.50	Initial value					
13	78.76	Simulated value, time T later					
14							
15	FORMULAS FROM RANGE A1:B13						
16	A5. =A2*A4						
17	A6. =A3*(A4^0.5)						
18	A8. =A7+NORMINV(RAND(),A5,A6)						
19	A13. =A12*EXP(NORMINV(RAND(),A5,A6))						

Figure 1. Simulating Brownian motion over a given time interval.

If we assumed that the price per share of stock changed over time according to a Brownian motion, then we would be assuming that the difference between the final price and the initial price during any year will be independent of the initial price at the beginning of the year. But if the price wandered down below $1 then a $1 price decrease would necessarily have probability 0, because negative prices are impossible! So we cannot realistically assume that prices change according to a Brownian motion.

But it can be reasonable to assume that, during any year, the ratio of the final price divided by the initial price will be independent of the initial price at the beginning of the year. This assumption is satisfied in a probability model where the natural logarithm of the price (rather than the price itself) changes according to a Brownian motion. In such a model, we say that the price is changing over time according to <u>log-Brownian motion</u>. In such a log-Brownian motion, the Brownian drift and volatility of the logarithm of the price may be called the <u>log-drift</u> and the <u>log-volatility</u> of the price.

Recall that the EXP function is the inverse of the natural logarithm function LN, and the EXP function converts addition to multiplication. So over some time interval, if the natural logarithm of a price changes by the addition of some quantity x, then the price itself over that interval is being multiplied by the quantity EXP(x). Thus, if a stock's price changes over time according to a log-Brownian motion with log-drift D and log-volatility V then, over any time inteval of length T, the price at the end of the inteval will be equal to the price at the beginning of the inteval multiplied by a Lognormal random variable of the form EXP(**X**), where **X** is a Normal random variable with mean D*T and standard deviation V*(T^.5).

Cells A12:A13 in Figure 1 show how to simulate such log-Brownian motion. When the initial value is entered into cell A12, the value after T years of log-Brownian motion is simulated in cell A13 by the formula

=A12*EXP(NORMINV(RAND(),A5,A6))

where A5 = A2*A4, A6 = A3*(A4^0.5), A2 is the log-drift, A3 is the log-volatility, and A4 is the length of the time interval T.

The log-drift and the log-volatility for such a log-Brownian motion model of price changes can be estimated from historical price data as shown in Figure 2. The range C7:C24 in Figure 5 computes the natural logarithms of past annual growth ratios of a stock price. Then the average and standard deviations of these logarithmic growth rates are computed in cells C1 and C2, to give us our estimate of the log-drift and log-volatility.

	A	B	C	D	E	F
1			0.1768	Log-drift		
2			0.3435	Log-volatility		
3	BOEING CO STOCK DATA					
4		Price per share				
5	Dec77	1.85	Log-growth rate	FORMULAS FROM RANGE A1:C24		
6	Dec78	4.70	0.931	C6. =LN(B6/B5)		
7	Dec79	5.00	0.062	C6 copied to C6:C24		
8	Dec80	6.54	0.268	C1. =AVERAGE(C5:C24)		
9	Dec81	3.33	-0.674	C2. =STDEV(C5:C24)		
10	Dec82	5.02	0.409			
11	Dec83	6.48	0.256			
12	Dec84	8.39	0.258			
13	Dec85	11.61	0.325			
14	Dec86	11.36	-0.022			
15	Dec87	8.22	-0.323			
16	Dec88	13.47	0.494			
17	Dec89	19.79	0.385			
18	Dec90	22.69	0.137			
19	Dec91	23.88	0.051			
20	Dec92	20.06	-0.174			
21	Dec93	21.63	0.075			
22	Dec94	23.50	0.083			
23	Dec95	39.19	0.511			
24	Dec96	53.25	0.307			

Figure 2. Estimating log-drift and log-volatility from historical data.

Figure 3 shows how such a log-Brownian motion model of an asset price can be applied to forecast the evolution of an asset price as well as the probability distribution of various derivative assets. We start with our estimated annual log-drift (0.1768) and log-volatility (0.3435) for the price per share of this stock, as entered into cells A3 and A4 of Figure 3. Let us also consider two derivative assets which, by definition, have values that derived from this stock price. The first is an option to buy one share of this stock 2 years from now for $60. The second derivative is an unconventional asset that promises to pay the highest monthly closing price per share of this stock over the next 2 years (24 months).

	A	B	C	D	E	F	G	H
1	LOG-BROWNIAN MODEL OF STOCK SHARE PRICE AND TWO DERIVATIVE ASSETS							
2	Given parameters of share-price process (estimated from annual data)							
3	0.1768	Log-drift (annual)						
4	0.3435	Log-volatility (annual)						
5	Calculated parameters for monthly simulation model							
6	0.0833	Period length (in years)						
7	0.0147	Mean logarthmic growth rate per period						
8	0.0992	Stdev of logarthmic growth rate per period						
9								
10	Month	Share value		Simulated final value (at 2 years) of an option				
11	0	54.50		to buy one share for $60 in 2 years.				
12	1	65.20		77.39				
13	2	70.81						
14	3	84.23		Simulated final value of a derivative paying,				
15	4	83.22		in 2 years, the highest monthly closing price				
16	5	86.54		per share of the stock over the next 24 months.				
17	6	98.76		148.69				
18	7	98.30						
19	8	106.91						
20	9	119.55						
21	10	144.77						
22	11	124.94						
23	12	126.19						
24	13	122.50						
25	14	109.61						
26	15	105.57						
27	16	109.32						
28	17	111.99						
29	18	131.00						
30	19	131.13						
31	20	127.85						
32	21	130.14						
33	22	148.69						
34	23	132.74						
35	24	137.39						
36								
37								
38								
39	FORMULAS							
40	A6. =1/12							
41	A7. =A3*A6							
42	A8. =A4*(A6^0.5)							
43	B12. =B11*EXP(NORMINV(RAND(),A7,A8))							
44	B12 copied to B12:B35							
45	D12. =MAX(B35-60,0)							
46	D17. =MAX(B12:B35)							



Figure 3.

To study the probability distributions of these assets, we need to develop a simulation model of the future prices per share of this stock for each of the next 24 months. One month is of course 1/12 of a year, and so the expected logarithmic growth rate for this share price over any one-month period is

0.1768*1/12 = 0.0147,

as computed in cell A7 of Figure 3. The standard deviation of the share's logarithmic growth rate over any one month period is

0.3435*((1/12)^0.5) = 0.0992,

as computed in cell A8. The current price per share of this stock is $54.50, which is entered into cell B11 of Figure 3. Then the simulated share prices at the end of each of the next 24 months are simulated in cells B12:B35 by entering the formula

=B11*EXP(NORMINV(RAND(),A7,A8))

into cell B12, and then copying cell B12 to B12:B35.

The final value of the stock 2 years from now is simulated in cell B35 of Figure 3. In two years, the option to buy one share for $60 will be either the share price minus $60, if this is positive, or will be $0 otherwise. So the simulated final value (at 2 years) of the option is computed in cell D12 by the formula

=MAX(B35−60,0)

which returns either B35−60 or 0, whichever is larger. The final value of the derivative asset that pays the highest end-of-month closing price of this stock over the next 24 months is simulated in cell D17 of Figure 3 by the formula

=MAX(B12:B35)

which returns the highest value among the simulated end-of-month share prices in B12:B35.

By making a table of hundreds of simulated values from cells B35, D12, and D17 in Figure 3, we can estimate the probability distributions of the values of this stock and these two derivative assets after 2 years. Figure 4 shows the inverse cumulative distributions for these unknown quantities, as estimated from a simulation table with 401 values.


Best Monthly Close
Share price
Option
Asset value in 2 years ($)
250
200
150
100
50
0
0
0.2
0.4
0.6
0.8
1
Cumulative probability

Figure 4.